Smalltalk ‘Traps’

This section discusses a few aspects of Smalltalk that often cause confusion. The examples given are far from exhaustive but they should alert you to some issues that warrant careful consideration.

A few of these examples should help to illustrate that writing Smalltalk can sometimes lead to unexpected results and occasionally create some serious problems.

Message Answers

Attempting to correctly determine the message answer returned after evaluating a Smalltalk expression is often difficult for those new to the language. Incorrect assumptions will lead to problems. 

Manipulating collection objects, in particular when adding objects to a collection, is a case in point. You might assume that the following expression would result in dateSet referencing an instance of Set.

dateSet := (Set new) add: hisBirthDate; add: herBirthDate.

This evaluates, but instead of dateSet referencing a set of Date instances, inspection will show that a Date instance has been assigned to the variable. The final message add: herBirthDate returns the argument and consequently this is assigned to dateSet. 

A solution that avoids this problem is:

dateSet := (Set new) add: hisBirthDate; add: herBirthDate; yourself.

The yourself message simply returns the receiver, the newly created instance of Set and consequently this is assigned to dateSet.
The following will also work:

dateSet := Set new.

dateSet add: hisBirthDate.

dateSet add: herBirthDate.

Or

dateSet := Set new.

dateSet add: hisBirthDate;

        add: herBirthDate.

The message answer in both of these examples is the last element added to the set i.e. the Date instance referenced by herBirthDate. However since no assignment is involved in the final line, dateSet still references an instance of Set.

Explicit message returns

The default message answer used by Smalltalk methods is to return the receiver and this will always be the case unless a message answer is explicitly returned by the use of the caret character ^. Smalltalk will return a value after evaluating an expression that immediately follows ^. Any further expressions will not be evaluated. 
This is a snippet of a possible method written for the Account class. 

You should assume that the specification states that if the user cancels a dialogue box, in response to either request, then the method should exit and return the receiver. 

getDetails

    "Prompts for a name and opening balance for the receiver ... the rest        of the comment has been omitted."

    |name openingBalance|

    name := Dialog request: 'Please enter name'

                  initialAnswer:  'A. Person'

                  onCancel: [Dialog warn: 'Account not properly set up'.                              ^self].

    openingBalance := Dialog request: 'Please enter opening balance'

                            initialAnswer: '100'

                            onCancel: [Dialog warn: 'Account not properly                                                               set up'.

                                       ^self].

    "The rest of the method body has been omitted."

Omitting the first explicit return (i.e. ^self) would mean that the method fails to exit and the second dialogue box still opens. Implementing this behaviour would not adhere to the specification.

It is possible to exit the method with

    name := Dialog request: 'Please enter name'

                   initialAnswer:  'A. Person'

                   onCancel: [^Dialog warn: 'Account not properly set up'.                          ].

However cancelling would return nil and not the receiver, again failing to follow the specification.

Precedence rules and the precedence tool

The Smalltalk precedence rules are: 

Bracketed expressions are evaluated first, work from left to right.

Unary expressions are evaluated left to right before (take precedence over) binary expressions; 

Binary expressions are evaluated left to right before (take precedence over) keyword expressions; 

Keyword expressions are evaluated last, again working left to right.

Although usually sufficient for most purposes, these rules occasionally lead us astray.

The precedence rules do not always accurately reflect the order of evaluation as implemented by the Smalltalk parser, the software that ‘reads’ the code. The parser works strictly left to right but before evaluating a message it looks ahead to the next message to check if it is of a higher precedence than the current message. If the next message is of higher precedence then it gets evaluated before the current message.

The precedence tool (LB- 14 Practical 14) uses the same rules as the parser and works well as long as brackets are not used. It works reliably for evaluating expressions that are combinations of binary, unary and keywords but you should avoid using it with bracketed expressions.

For example:

aFrog right position:(aFrog position)

If you follow the precedence rules you will evaluate the bracketed message first. This will lead you to a wrong evaluation. The parser evaluates aFrog right first (it reads left to right), this is the first unary message. If aFrog is a newly created instance with position of 1 then the above expression will set its position to 2 not 1

Another example:

10 + (10 negated) negated

The precedence tool finds + realises it must look ahead one for a message of higher precedence. In the look ahead it instead finds the brackets and correctly evaluates the contents of those brackets. It then ‘forgets’ that it still hasn't completed the original search for a message of higher precedence and incorrectly evaluates the binary message (10 + -10) before sending negated to 0. The tool leads you to an incorrect answer of 0. The correct result of the message expression should of course be 20.

Students new to writing Smalltalk will find that the careful use of brackets will help ensure that evaluation is completed successfully as intended..

For example: 

kermit position: 2 green 

Error! Precedence again. Smalltalk attempts to evaluate the unary message 2 green first. 
(kermit position: 2) green 

Evaluates. Smalltalk evaluates the bracketed expression first.

Here is another common error.

ouAccount debit: 300 holder: 'The Open University'

But the unthinking use of brackets will cause further trouble

(ouAccount debit: 300) holder: 'The Open University'

Cascading will solve this problem.

ouAccount debit: 300;

          holder: 'The Open University'

Care is needed with fractions.

3/4 denominator "answers 3"

3/4 numerator "answers 3/4"

You need to write

(3/4) denominator. "answers 4"

(3/4) numerator    "answers 3"

3/4 is a message expression that answers with an instance of Fraction it is not a literal.

Weird Smalltalk 

Smalltalk is written in a spirit of cooperation and openness. Convention states that the name used as the argument for a method should suggest the class of the object supplied. But it is only a suggestion and enforcement is often not implemented. The user of a method is responsible for ensuring that any data supplied accords with the specification given in the initial comments.
harry := HoverFrog new.

harry hover: Green by: 2.

harry height  "answers 2 because if the argument to hover: is not Down it is assumed to be Up"

m206Account := Account new balance: Frog.

m206Account balance "answers Frog – again no data validation"
A string conundrum

1 to: 2 do: [ :i |



croaker := 'cat'.



(i = 1) ifTrue: [ croaker at: 1 put: $r ].



Dialog warn: 'Croaker is a ', croaker]

At first sight this innocent looking snippet of Smalltalk appears to be straightforward. Yes, it’s going to loop twice, but the result may be unexpected. 

On the first loop croaker references the same object as the string literal 'cat', so when we change croaker, we actually change the object that the literal references.

When Smalltalk executes, it does not reset literals according to what is in the source code, which is what you would probably expect from the second loop through. What we get is:

This is fine for immutable objects like integers or symbols (unless one attempts some of the code to be found in the next section), but a trap for literals creating mutable objects like strings.

The solution to avoid this?

1 to: 2 do: [ :i |



croaker := 'cat' copy.


(i = 1) ifTrue: [ croaker at: 1 put: $r ].



Dialog warn: 'Croaker is a ', croaker]
So croaker references a new copy of the object which can be changed without corrupting the literal. Using copy, on the second loop through we get:

Sorted collections

deadPoets := SortedCollection new.

deadPoets add: 'Milton'; add: 'Shakespeare';  add: 'Keats'.

The textual representation of the sorted collection shows that the strings have been sorted as expected.

SortedCollection ('Keats' 'Milton' 'Shakespeare')

Now evaluate the next two lines

jk := deadPoets at: 1.

jk at: 1 put: $Y.

The textual representation of the sorted collection is now

SortedCollection ('Yeats' 'Milton' 'Shakespeare')

Sorted collections do not sort themselves if manipulation of the stored objects takes place after they have been added. So we need one final line of code:

deadPoets resort

Modifying a collection with do: 

Although not applicable to the breeding patterns of frogs, Fibonacci numbers are reflected in many aspects of the natural world. Beginning with 0 and 1, a series of numbers can be formulated with each successive number being the sum of its two predecessors. The first twenty numbers in the series are

0 1 1 2 3 5 8 13 21 34 55 89 144 233 377 610 987 1597 2584 4181 

A Smalltalk Fibonacci number generator can be built with the following:

fibonacci := #( 0 1) asOrderedCollection.

18 timesRepeat: [ fibonacci add: ((fibonacci at: fibonacci size - 1) + fibonacci last)]
Now assume that we wish to remove all of the odd numbers in this collection.

fibonacci do: [ :f |

    (f \\ 2 = 0) ifFalse: [fibonacci remove: f ] ].
You will quickly discover that this will fail. The do: message is modifying the collection whilst iterating through each of its elements.

A neat solution once again involves the use of copy.

fibonacci copy do: [ :f |

    (f \\ 2 = 0) ifFalse: [fibonacci remove: f ] ].

The copy of the collection is iterated through but, unlike the original, it is not modified.

Dangerous Smalltalk

It is quite simple to create havoc in a LB. 

Under no circumstances evaluate any of this code and then attempt to save your Learning Book.

For example:

Red := Green. "Global variable Red now references Green!"

hoverFrog1 := HoverFrog new.

hoverFrog1 colour: Red.

hoverFrog1 colour "answers Green"

Or this:

Frog := HoverFrog new. "Danger!"

Frog class == Frog "answers false"

M206 teaches you that symbols are immutable but Smalltalk hides the reality and attempting to prove otherwise can cause considerable problems. 

Here is a good example from a Course Team member. He writes:

“I once managed to change the 'immutable' symbol #right so that when evaluated it answered #wrong. You should see what that did to my frogs.”

So how did he manage it?

A LB with a World page was used and a Frog instance called kermit created. Then the following was evaluated in the Workspace:

#right 


basicAt: 1 put: 119;


basicAt: 2 put: 114;


basicAt: 3 put: 111;


basicAt: 4 put: 110;


basicAt: 5 put: 103.

#right

This answers with the symbol #wrong.

Evaluating #right again returns #right.

In the GUI clicking the right button works as it should. 

However evaluating

    kermit right 

produces this 

kermit wrong 

produces

[image: image7.png][ [x]

wiong is a new message





In the class browser you would observe that Frog now has an instance method called wrong but no right method. But if you click on the wrong method you will get this

A final warning from the author of this code:

“I'm not sure I can remember how to get out of this mess but would suggest not saving the LB until you find out.”

